```
var Y C I N K R W A G theta y c i w r n g lt a;

varexo eA et eG;

parameters beta alpha delta chi gs ggs thetas rhoA rhot rhoG sA st sG ;

load param_rbc;
set_param_value('alpha',alpha);
set_param_value('beta',beta);
set_param_value('delta',delta);
set_param_value('chi',chi);
set_param_value('gs',gs);
set_param_value('ggs',ggs);
set_param_value('thetas',thetas);
set_param_value('rhoA',rhoA);
set_param_value('rhot',rhot);
set_param_value('rhoG',rhoG);
set_param_value('sA',sA);
set_param_value('st',st);
set_param_value('sG',sG);

model;

% (1) Euler equation capital
1/C = beta*(1/C(+1))*(R(+1) + (1-delta));

% (2) Labor supply condition
theta*N^(chi) = (1/C)*W;

% (3) Labor demand
W = (1-alpha)*A*K(-1)^(alpha)*N^(-alpha);

% (4) Capital demand
R = alpha*A*K(-1)^(alpha-1)*N^(1-alpha);

% (5) Production function
Y = A*K(-1)^(alpha)*N^(1-alpha);

% (6) Capital accumulation
K = I + (1-delta)*K(-1);

% (7) Resource constraint
Y = C + I + G;

% (8) Process for A
log(A) = rhoA*log(A(-1)) + sA*eA;

% (9) Process for G
log(G) = (1-rhoG)*log(ggs) + rhoG*log(G(-1)) + sG*eG;

% (10) Process for theta
log(theta) = (1-rhot)*log(thetas) + rhot*log(theta(-1)) + st*et;

% (11) log Y
y = log(Y);

% (12) log C
c = log(C);

% (13) log I
i = log(I);

% (14) log W
w = log(W);

% (15) log R
r = log(R);

% (16) log N
```

```
n = log(N);

% (17) log G
g = log(G);

% (18) log theta
lt = log(theta);

% (19) log A
a = log(A);

end;

steady;

shocks;
var eA = 1;
var eG = 1;
var et = 1;
end;

stoch_simul(order=1,irf=20,nograph,ar=1);
```